Department of Physics and Astronomy

Meteorology Experiments Fourth Year Project

Chaos and Weather Forecasting : Experiments with the ‘Lorenz’ model

Peter Ruszczynski March 1996

Abstract

Throughout the ages, has shain a special interest in predicting the weather. However the methods of forecasting have changed over this time. In this paper I investigate a system of three coupled differential equations of first order, first introduced by E. N. Lorenz in 1963. This system, known as the ‘Lorenz model’, describes fluid convection and can be applied to the dynamics of the atmosphere. It describes that the solutions of the system can have very different types of behaviour as the parameters of the model change. One of them is the well known phenomena of Chaos which causes difficulties for weather prediction. In the second part, I present an extension of the model which considers a forcing from the tropical atmosphere and it’s effect on climate forecasting.

Supervisor: Dr. Keith Haines 6 Weeks

Contents

1 Introduction 1

2 The Traditional Lorenz Model 2 2.1 The Equations . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 2 2.1.1 The Variables and Parameters . . . . . . . . . . . . . . . . . . 3 2.1.2 Global Properties of the System . . . . . . . . . . . . . . . . . 4 2.2 The Behaviour of the Solutions . . . . . . . . . . . . . . . . . . . . . 4

3 Extensions of the Model 9 3.1 Mathematical Extensions and their Meaning in Meteorology . . . . . 9 3.2 The Behaviour of the Solutions . . . . . . . . . . . . . . . . . . . . . 10

4 Conclusion 12 4.1 The Traditional Lorenz Model . . . . . . . . . . . . . . . . . . . . . . 12 4.2 The Extensions of the Lorenz Model . . . . . . . . . . . . . . . . . . 14

5 Used Software 15

6 References 15

A Appendices 16 A.1 Source text : lorenz.f . . . . . . . . . . . . . . . . . . . . . . . . . . . 16

1 Introduction

Meteorologists use mathematical models, developed from basic physical laws to predict the weather of tomorrow. Basic physical laws for example are Newtons laws, from which the so called Navier Stokes Equation, describing the dynamics of a fluid , say atmosphere, can be derived . The Navier Stokes equation together with the thermal diffusion equation can be used to build up the so called Lorenz model. The Lorenz model which consists of three coupled, ordinary differential equations of first order was introduced by Edward N. Lorenz in 1963. Since he published in the ‘ Journal of the Atmospheric Sciences ’, his paper gained recognition by the physicists quite late on. It was one of the reasons for the boom of the Chaos Theory in the late 70’s and 80’s. Lorenz showed that the solutions ( the motion of the variables of the system in time ) can have a behaviour which we call chaotic because of the sensitivity of the solutions to initial conditions.

The solutions can be observed in the state space, that is the three dimensional space with the three variables of the model as the basis. In the case of the chaotic motion he obtained a quite strange object in that space into which all trajectories will move in, within a finite time. This object is called a strange attractor. Since this strange attractor looks like the two wings of a butterfly the effect of sensitivity to initial conditions was later called the ‘Butterfly effect’. A famous saying was born : ‘The flapping of the butterfly’s wings anywhere in the world can cause a tornado anywhere far away.’ This together with some nice pictures of fractals made the chaos theory popular. The unpredictability of everything sounded quite interesting and suggested to people that they could spend the time they are watching the weather report in a better way. But the boom is almost over since the unpredictability of everything has not been confirmed.

The mathematical reason for this is the following: Even if the equations are nonlinear, the nature can often be described by a linearised, that is, truncated, version, of these equations. This is possible so that the nonlinear terms which can cause chaos have no effect to the behaviour of the system. A prediction for the weather, the so called prediction of the first kind, is possible if the influence of the nonlinearity can be neglected and rather difficult otherwise. Why is shown in the first part of this paper.

The long time behaviour of the Lorenz model, represented by mean values, is almost independent on the initial conditions. We can use that to define what is long time behaviour, but one has to be careful to apply that to the atmosphere, because the connection between the integration time units of the Lorenz model and the time scale for the dynamics of the atmosphere is not yet clear. But assuming that the long time bahaviour of the system corresponds to the long time behaviour of the atmosphere, the climate of the future can also be studied by the equations. This is the so called prediction of the second kind. This kind of prediction is also limited, but the reasons are different ones: The problem of climate forecasting is not caused by the problem of an inaccurate knowledge of the initial conditions, but by the fact that we don’t know all the parameters and their values which will determine our climate in the distant future, say 20 or 30 years. Another problem is that the Lorenz model does not gives us a complete description of the atmosphere.

In the second part, I extend the model with an additional term as T. N. Palmer did in 1993 [3] and discuss results which are relevant especially for the predictions of the second kind.

2 The Traditional Lorenz Model

2.1 The Equations

The Lorenz model is the following closed system of three, coupled, ordinary differential equations of first order: Ẋ = −σX + σY Ẏ = −XZ + RX − Y Ż = XY − bZ They can be derived from the Navier-Stokes equation for fluid flow and the thermal diffusion equation with a lot of truncations. A development of these equation is given in [1].

The variables are not spatial variables and their meaning is given later. The Lorenz model describes a two dimensional system which has been studied experimentally by Bénard in 1900: a layer of an incompressible fluid is heated at the bottom and cooled at the top. Considered are only the vertical and one of the horizontal dimensions. Since Lord Rayleigh based that experiment theoretically, the system is called a Rayleigh-Bénard cell. The control parameter of the system was chosen to be the temperature difference between the bottom and the top, say δT = T_b − T_t. One can find that the fluid shows very different types of behaviour as δT is increased from zero:

For very small δT , just greater as zero, a warmer fluid ‘packet’ at the bottom will tend to drift upwards but loose all the thermal energy on its way because of the viscosity of the fluid. It never reachs the top and never causes convections. As δT passes a critical value the fluid ‘packet’ will reach the top, convective motion will set in and the system will start to build up a pattern. This pattern consists of convection rolls and is caused by the rising and falling fluid ‘packets’. If we assume a reposing fluid at t = 0 then the direction of the turns (clockwise or counterclockwise) at t > 0 will be determined by very small fluctuations, say Brownian motion. The intensity and shape of both possibilities for the convection rolls is identical, except for the sign.

Increasing δT further leads to a second critical point and the third type of behaviour: turbulent motion.

All of these kinds of behaviour are covered by the three equations of the Lorenz model.

The application of the model in meteorology is justified by treating the Earth’s atmosphere as a fluid layer which is heated at the bottom due to the suns heating of the Earth’s surface.

2.1.1 The Variables and Parameters

There are three variables and three parameters in the system of differential equations. The meaning of them in physical terms are the following:

Variables:

X : is related to the time derivation of the stream function of the fluid, or more simply put it describes the intensity of the convective motion. So X can be treated as a measure of the rate of convective overturnings.

Y : is proportional to the temperature difference of the ascending and descending current of fluid at a given height.

Z : measures the temperature difference between the actual temperature profile and the linear temperature profile.

If the signs of X and Y are the same it means that the warm fluid ‘packets’ are ascending and the cold are descending.

Parameters:

σ : is the Prandtl number which is the ratio of the kinetic viscosity and the thermal diffusion coefficient of the fluid, so that σ compares the energy lost by friction and by thermal conduction.

R : is proportional to the Rayleigh number which is itself proportional to the temperature difference δT between the bottom and the top of the fluid layer. b : is a geometrical parameter which is related to the ratio of the layer height and the horizontal size of the convection roll.

All the parameters are assumed to be positive.

2.1.2 Global Properties of the System

The most important property is that this system of differential equations is not analytically solvable and a numerical investigation is necessary. But it is possible to analyse some features of the system ‘by hand’:

The system of equations is symmetric under the transformation (X, Y, Z) → (−X, −Y, Z) . That leads to symmetric solutions for X and Y . As a second property one can show that there is a region in the state space which the trajectory ( the path of the solution in time ) will never leave if it has entered this region once. It means that there exists a boundedness of the solutions for all times.

Another global property is that a trajectory starting on the Z-axis will move into the origin (0, 0, 0) .

2.2 The Behaviour of the Solutions

There are three different types of behaviour depending on the values for the parameters in the parameter range I discuss here. That means that there are two points where the behaviour of the system changes. It is possible to calculate these points:

Solving the stationary system 0 = −σX + σY 0 = −XZ + RX − Y 0 = XY − bZ gives three equilibria points (X_1, Y_1, Z_1) = (0, 0, 0) and (X_{2/3}, Y_{2/3}, Z_{2/3}) = (± √ b(r − 1), ± √ b(r − 1), r − 1) . If the Lorenz model is written as ˙⃗x = f(⃗x) with ⃗x = (X, Y, Z) and matrix f(⃗x) then the matrix of the derivations is

Df(X, Y, Z) = −σ σ 0 R − Z −1 −X Y X −b The linearised matrix around the orign is then: −σ σ 0 R −1 0 0 0 −b The (X, Y ) equations have a trace τ = −σ − 1 and a determinant △ = σ(1 − r) , so that for R < 1 the orign is attracting (negative trace and positive determinant). If R > 1, one eigenvalue of the matrix is positive and one is negative, so that the origin is a saddle point and not anymore a purely attracting one. That means that R = R_1 = 1 is the first point where a changing of the behaviour of the system is expected.

Writing down the matrix Df(X, Y, Z) for the second (and similarly for the third) equilibrium point (say C and C0) leads to the linearised matrix : −σ σ 0 1 −1 −X X X −b The characteristic polynomial for the eigenvalues λ follows as

λ^3 + (σ + b + 1)λ^2 + b(σ + R)λ + 2σb(R − 1) = 0 ,

so that the eigenvalues are the roots of the cubic equation

λ^3 + kλ^2 + lλ + m = 0 with k = σ + b + 1, l = b(σ + R) and m = 2σb(R − 1).

If R > 1 all the eigenvalue are negative if m < kl , i.e. R <^{σ(σ+b+3)}_σ�_b�_1 is the condition for C and C0 to be attracting. For the parameter σ = 10 and b = 8/3, which are mostly chosen to investigate the Lorenz attractor, the value for R becomes 24.74.. .

The condition for the cubic equation to have a purely imaginary pair of roots is l > 0 and m = kl. Since the assumption gives l > 0, this leads to R = R_2 =^{σ(σ+b+3)}_σ�_b�_1 . That is a value for which a Hopf bifurcation occurs if σ > b + 1 and the second point where the behaviour of the system changes. It can be shown that the bifurcation is a subcritical one, which means that the fixed points C and C0 become unstable. For greater values then R_2 a numerical investigation is necessary. To summarise these analytical results it is worth to look to Fig.1.

Figure 1: The plot of the fixed points for various values of R; (σ = 10, b = 8/3) shows the three different types of behaviour.

Plotted are values for X after the systems had enough time to evolve to its characteristic behaviour. The parameter for σ and b are fixed for this figure as well as for all further investigations: (σ = 10, b = 8/3). The two points where the behaviour of the system changes are easy to verify. For R > R_1 = 1 the origin is not attracting anymore it becomes unstable and the points C and C0 become stable. The orign corresponds to the state of no convection (no overturnings, no temperature deviations from the linear profile) the values into which the system evolves for R > 1 are correlated with steady convection. Easy to verify is the symmetry for each value of R. At R = R_2 the situation changes again, every point seems now to be possible for the system to evolve in. This corresponds to turbulent motion.

The first two states (0 < R < R_1) and (R_1 < R < R_2) are stable. This means that a meteorological prediction for the behaviour of the atmosphere is possible for all times if the equations cover all the dynamics of the atmosphere. But one has to point out that these equations are only a very rough truncation of the more fundamental Navier-stokes equation, and are far from describing the atmosphere completely. But nevertheless the Lorenz model is derived from them and the properties of the solutions can be expected to occur in the dynamics of the atmosphere.

Figure 2: The strange attractor of the Lorenz model for the parameters (σ = 10, R = 28, b = 8/3), where ‘chaos’ occurs.

Of special interest is the phenomena of chaos (R > R_2):

If the parameter R exceeds the value of R_2 the two fixed points C and C0 become unstable. If one plots the trajectory in the state space for times of sufficient duration, i.e. to observe the solution of the system, then one finds an attractor as shown in figure 2.

This is the famous Lorenz attractor, a so called strange attractor because of the regularity, where one would expect a rather random shape. The trajectory never crosses itself and never repeats a loop it has done before. The property which is called chaotic is the property that nearby trajectories tend to diverge exponentially, so long as the bounded region allows that. This means that a little uncertainess in the initial conditions can make a prediction rather impossible. An example is shown in figure 3 where two trajectories start nearby. The integration time which exhibits such a divergence is only 3.8 time units, and the initial conditions are given in the figure. In fact one can point out that the more nearer two trajectories are at the starting point, the more time it will need to observe the maximal distance between them given by the size of the bounded region or in other words the motion of the two trajectories will remain similar for longer. This means that a prediction can be made more accurate if the uncertainness is smaller.

Figure 3: The exponential divergence within the strange attractor. Both trajectories start inside the circle.

If a trajectory comes very close to the orign (which it can never reach exactly, starting outside the Z-axis, because the orign is a repelling fixed point) the trajectory tends to move on the edge of the attractor for a while. The closer the trajectory comes to the orign, the longer it will remain on the edge of the attractor. The trajectory, indicated by the solid line in figure 3 is such an example. The separation of both trajectories takes place very close to the Z-axis (X = 0, Y = 0) inside a region where the motion is especially sensitive to little perturbations. Unfortunately that is the region where the trajectories spend most of their time, as a plot of the probability distribution function (PDF) in figure 4 shows. The peak at (X = 0, Z ≈ 30) shows that the trajectory is mostly expected here. This is the region where the two ‘wings’ touch and the trajectory may tend to the right or left ‘wing’ depending on the initial conditions and eventually small perturbations. Note that the regions around C and C0 looks equal.

A trajectory in the Lorenz attractor moves from one ‘wing’ to the other af-

Figure 4: The PDF in the X −Z plane for the chaotic parameter regime. The higher the ‘mountain’, the bigger is the probability to find the trajectory here.

ter spending some time spiralling round about one of the unstable fixed points C or C0. This leads to a quasiperiodic motion with a frequency of approximately 1.5 timeunits�^1. One can verify this in figure 5 . The example in figure 5 is characteristic for the behaviour of a trajectory inside the Lorenz attractor: coming, for example, from the negative region^1 , spiraling inwards and/or outwards round about the positive fixed point and turning back to the negative region to spiral there. The number of loops of such spirals is very sensitive to the initial conditions, i.e. if we find for example a 10-loop spiral at t ≈ 1460 we must not find one there if we slightly change the initial conditions. Since the Lorenz equations are symmetric it is clear that the solutions must be symmetric. More accurately this holds only for X and Y , i.e. the trajectory must spend the same time in the positive region as in the negative region as t → ∞. It is obviously that the mean values for X and Y must be 0 as t → ∞.

If we indicate the revolvings round about a fixed point with regime like behaviour of the atmosphere we can point out that if someone had found a periodicity at the given parameters (σ, R, b) one can expect the time which this positive or negative regime remains, say the regime existence time (RET), to be a whole-numbered multiplicatively of the Lorenz period T_L ≈ 0.67. Unfortunately since it is impossible

1Negative region shall be the sub-state-space where X < 0 and Y < 0; positive region

respectively.

Figure 5: Quasiperiodic motion inside the strange attractor which corresponds to a spiral outwards from the positive fixed point

to know the exact initial condition one will not know the whole number, i.e the number of the loops.

But one can find out that short RET’s occur more often then long RET’s. This seems to be according to an exponential law as figure 6 indicates. To produce figure 6, I have counted how often a certain time interval (RET) occur where X is positive for the time t = 10^3 to 10^5. The peak like character of that graph is caused by the Lorenz period T_L : For a RET ≈ 10 time units we have 15 peaks and^{10}_{15} ≈ T_L ). That confirms what I have stated above. Whole-numbers multiplicatively of T_L are the most probably times which a trajectory remains in one of the regions. The picture for Y would look the same as well as the picture for the negative region because of the exact symmetric model for X and Y .

In summarising one can say that even a motion inside a strange attractor can have some properties and a prediction can be based on them.

3 Extensions of the Model

3.1 Mathematical Extensions and their Meaning in Meteorology

An extension of the Lorenz model can be made by adding a forcing vector ⃗F = (F_X, F_Y , F_Z) to the equations. The system then looks as follows: Ẋ = −σX + σY + F_X Ẏ = −XZ + RX − Y + F_Y Ż = XY − bZ + F_Z

Figure 6: The frequency of the times which the trajectory stays inside the region X > 0 of the Lorenz attractor. The peaks are related to the Lorenz period T_L ≈ 0.67 time units. The RET, corresponding to the first peak, occured 505 times.

These additional terms can be understood as a forcing from the earth’s tropics and has been discussed by T. N. Palmer in 1993 [3]). The forcing term causes the system to be not anymore symmetric in X and Y .

The effect of the forcing to the Lorenz attractor can be qualitatively seen by looking to the equations now. If we assume ⃗F to be positive (otherwise we would have a forcing from the North Pole) the time deviations increases more and the state vector is expected to tend more to the positive region.

In particular we assume as T. N. Palmer a forcing vector ⃗F = (F_X , F_Y , 0) so that F_Z = 0. This means that the rate of positive convection should be increased as well as the temperature differences between rising and falling parts of the atmosphere, i.e. overall clockwise convection.

3.2 The Behaviour of the Solutions

If we look now again to the PDF (in figure 7) we indeed obtain a shifting of the PDF to the positive region. It is easy to verify that the probability has changed as F_X and F_Y are increased from 0. This is happening without a changing of the points C and C0, so that the attractor looks the same. The projection on the top of figure 7 shows clearly the distortion of the symmetry.

To quantify the changes of the PDF one can count the positive and negative

Figure 7: The PDF in the X − Z plane for F_X = F_Y = 3.8. The region for positive X is more probably because of the forcing.

‘events’ over a long enough time and plot the ratio^P �^N_{P +N} where P is the number of the positive and N the number of the negative ‘events’. In the case F_X = F_Y = F_Z = 0 this ratio should be always close to zero and deviate only for small times according to the loops in either of both regions. Thats why it is necessary to take the mean value over a wide range of time.

For F_X = F_Y > 0 the ratio is greater then zero. This is shown in figure 8 for F_X = F_Y = 3.0 comparing with the results for F_X = F_Y = 0. Plotted is the ratio for 10, 000 timeunits. Until time t ≈ 2, 000 the curves are much distorted by the first loops of the trajectory. Afterwards it converges to the expected values. Since the ratio^P �^N_{P +N} is a measure for the preponderance of P , the ratio^P_{P +N} is a measure for the probability to find a positive ‘event’. The dependence of this probability on the forcing vector ⃗F is shown in figure 9 and looks almost linear. Plotted are the probabilities (as an average over 10000 time units) for each value of F_X = F_Y after convergence. One has to take the average because the ratio^P_{P +N} can vary locally, caused by an actual spiral in one region. Exact values for the probabilities one gets again only for an average for t → ∞. Interesting in figure 9 is the point at F_X = F_Y ≈ 4.2. The system becomes stable here and the probability to find a state vector with positive X-value is 1. This is independent on the initial conditions:

Figure 8: The ratio^P �^N_{P +N} is a measure of the preponderance of positive values and is not anymore 0 in the case of a forcing.

The system evolves to a fixed point inside the positive region. This means steady convection.

The indicated rangesin figure 9 at F_X = 1, 2, 3 and 4 show the standard deviations for each probability value. They are smaller then the raw curve allows. The reason for this are long RET’s which can occur in long times. This illustrates the difficulties for the predictions of the second kind: The number of loops and the time they occur is very sensitive to the initial conditions and can affect also results for large times. This is especially relevant to the forced system because large RET’s occur here more often then in the unforced system, as the plot in figure 10 shows when compared with that of figure 6. The frequency of shorter spirals decreases as the frequency for longer spirals increases.

4 Conclusion

4.1 The Traditional Lorenz Model

The application of the Lorenz model to the Earth’s atmosphere is possible in so far as it describes a quite basic system. A difficulty causes the relation between the time scale of changes of the atmosphere and the integration time in the Lorenz model. How many time units of the integrations correspond to times for predictions of the first kind and how many time units has to be considered to investigate predictions of the second kind ? A very careful identification can be made by the subdivision into phenomena which depend ‘very’ sensitive on initial conditions and phenomena which do not, let’s say phenomena of the first and second kind. Atmospherical phenomena of the first kind would then correspond to integration times like 1 to 100 time units, phenomena of the second kind to integration times like 10^3 to 10^5 time units. Assuming this relation, conclusions for predictions can be reached from the results of this paper if we have identified the parameter with measured values of the atmosphere.

Phenomena of the first kind are impossible to predict exactly in the chaotic param-

Figure 9: The Probability to find ‘positive event’s’ increases as the forcing vector becomes larger. At F_X ≈ 4.2 the system becomes stable

eter range. The sensitivity to initial conditions is not to deny as well as an unexact knowledge of the initial conditions which we will always have. Even if we would know the initial conditions mathematically exact we have no possibility to calculate the trajectory exactly , i.e. to predict the behaviour in the future, since the computer works with rounding errors. As we have seen in section 2, a little uncertainty grows exponentially and a rounding error is one. But restrictions for this unpredictability can be made:

In the case of no chaotic motion we can predict the behaviour for all times since we have calculated the fixed points and know how they change with the parameter. In the case of chaotic motion there are regions in the state space where the behaviour is quite regular. These are the regions around the two centroids of the ‘wings’. Once entering this region, the trajectory starts to build up a spiral round about a centroid for which we know the period and an exponential law for the frequency distribution of the existence time for such a regime.

Chaotic motion occurs for a parameter greater than a critical value. Recalling the meaning of the parameter, R is proportional to the temperature difference between top and bottom of the fluid layer or the atmosphere. If we consider the global increase of the temperature as measured between the years 1950 and 1992 (IPCC update 1992) and know ‘which R’ we have now, then predictions when we will pass the value R_2 might be possible, assuming we know the time evolution of R for the future or can extrapolate it. We could then predict when the difficulties of our predictions become a new quality.

Figure 10: The frequency of the RET’s. The comparison with figure 6 shows that the forcing causes longer RET’s

4.2 The Extensions of the Lorenz Model

Almost all conclusions from above are valid for the forced system since the attractor has not changed dramatically.

The most important conclusions for the forced system itself is with sure the change of the probability distribution function PDF. Positive values for the intensity of convections and the phenomena of ascending warm and descending cold fluid or atmosphere ‘packets’ occur more often in the statistical average. The PDF is almost linear to the strength of the forcing vector. This affects phenomena of the first kind as well as phenomena of the second kind. Trajectories which are in the most critical region around the Z-axis will tend more often to enter the positive region. That causes a change in the frequency of the RET. The RET’s become larger as in the unforced system, i.e. a periodic clockwise convection will remain longer as a counterclockwise one in the statistical average.

Another conclusion can be reached. If we assume a forcing vector whose strength vary with time and consider the almost linear relation to the PDF we could predict the variations for the PDF. Investigations of nonconstant forcing vectors has been made, for example in Molteni et. al. 1993.

5 Used Software

To integrate the Lorenz model I wrote Computer programs in Fortran 77. The programs use a standard library integration packet. In particular I used a 4-th order Runge-Kutta method which is much more accurate than the method Lorenz used in 1963 because of a truncation error of only order 5. Additionally, the programs run with adaptive step size control. This means that the step size is increased or decreased, depending on the comparison of a calculated error and a user defined tolerance. This error is calculated by the difference between the results of the calculation of the next step with step size h and the results calculated by 2 steps of step size h/2. This lets in fact the program run faster. The program uses double precision real numbers.

My first step was to reproduce the Lorenz attractor to control the correctness of the program. I had chosen the same parameter as Lorenz did in 1963: (σ = 10, R = 28, b = 8/3). I produced quite accurate pictures from the attractor in all three plane with ‘ xmgr ’. Unfortunately it is impossible to produce 3 dim pictures with it. ‘ graph2d -3 ’ can produce 3 dimensional plots but only with a limited number of some hundred points only.

‘gnuplot’ is also quite primitive but can handle larger data files, but it is unfortunately unable to produce reasonable hardcopies.

To produce the 3 dimensional image (figure 2) I used ‘ maple ’, which solves the Lorenz system itself and can produce plots! Unfortunately, the hardcopy version of a doted attractor is not anymore a dotted one but one with a lot of little crosses so that one can not see anything clearly.

Then I wrote a variation of my program to plot the probability in (X, Y )−, (X, Z)− and (Y, Z)−plane and in the (X,Y,Z) space. To visualise them I used the software packet UNIMAP which can produce 4-dim plots of the probability with the fourth dimension as a certain colour. Figure 4 and 7 are produced with UNIMAP. Unfortunately, UNIMAP is unable to produce a hardcopy version of the presentation window (as is promised in the guide). The hardcopy contains only the information of the left 2/3 of the presentation window, so that I had to delete the legend which would show a grey scale with the corresponding values.

Later on I wrote some simple parameter loops to produce datas such as for figure 1 and 9 and visualised them with ‘ xmgr ’ which works quit nice and is really recommendable !

6 References

(1) Hilborn, R.C.; 1994; ‘Chaos and Nonlinear Dynamics’; Oxford University Press (2) Lorenz, E.N.; 1963; ‘Deterministic Nonperiodic Flow’; Journal of the Atmospheric Sciences; 20; pp. 130-141 (3) Palmer, T.N.; 1993; ‘Nonlinear Dynamical Perspective on a Climate Change’; Weather; Vol.48;No.10;pp.314-326 (4) Sparrow, C.; 1982; ‘The Lorenz Equations: Bifurcations, Chaos, and Strange Attractors’;Springer-Verlag (5) Verhulst, F.; 1990; ;‘Nonlinear Differential Equations and Dynamical Systems’; Springer-Verlag

A Appendices

The following source text for the main version of the computer program, which integrated the Lorenz model, is written in Fortran 77 and uses the standard library Runge-Kutta order 4 method. The source text is commented.

A.1 Source text : lorenz.f

C-------------------------------------------------------------------------
      PROGRAM LORENZ

      IMPLICIT NONE

      REAL*8 So(3), S(3), EST1(3), EST2(3), EST(3), DSDT1(9), DSDT2(9)
      REAL*8 H, H_NEW, T, T_COPY, T_UP, T_END
      REAL*8 TOLERANCE, ERROR
      REAL*8 SIGMA, R, B, AX, AY, AZ !parameter: Ai = forcing
      REAL*8 COUNTER, T_PRINT, NR_DATA
      INTEGER STP, STP_UP
      CHARACTER LOG1, LOG2, LOG3, LOG4, LOG5
      LOGICAL LOG_A

      COMMON SIGMA, R, B, AX, AY, AZ

      EXTERNAL RHS
      EXTERNAL DRKSTP

C------------------------ INPUT : ----------------------------------

      WRITE(*,*)’__________________________________________’
      WRITE(*,*)’ENTER TOLERANCE FOR ADAPTIVE STEPSIZE :’
      READ(*,*) TOLERANCE
      WRITE(*,*)’ENTER OVERALL INTEGRATION TIME ( T_END ) :’
      READ(*,*) T_END
      WRITE(*,*)’ENTER TIME ( T_UP ) UP WHICH DATA SHALL’
      WRITE(*,*)’BE PRINTED TO THE DATA FILE(S) :’
      READ(*,*) T_UP
      WRITE(*,*)’ENTER NUMBER ( NR_DATA ) OF DATA POINTS’
      WRITE(*,*)’PRINTED TO FILE :’
      READ(*,*) NR_DATA
      WRITE(*,*)’ENTER INITIAL Xo :’
      READ(*,*) S(1)
      WRITE(*,*)’ENTER INITIAL Yo :’
      READ(*,*) S(2)

      WRITE(*,*)’ENTER INITIAL Zo :’
      READ(*,*) S(3)
C WRITE(*,*)’ENTER PARAMETER "SIGMA" :’
C READ(*,*) SIGMA
      WRITE(*,*)’ENTER PARAMETER "R" :’
      READ(*,*) R
C WRITE(*,*)’ENTER PARAMETER "B" :’
C READ(*,*) B
      WRITE(*,*)’ENTER AX, AY, AZ, ( I.E. FORCING) :’
      READ(*,*) AX
      READ(*,*) AY
      READ(*,*) AZ

      SIGMA = 10
          B = 2.666666666666666666

C-------------------- SOME VALUES : ---------------------

            H = 1.0d-10
            T = 0.0d0
          STP = 0
      T_PRINT = ( (T_END - T_UP) / NR_DATA )
      COUNTER = 0.0D0
        LOG_A = .TRUE.
           So = S

C----------- DATA TO FILE TO BE CHOSEN : ------------------

      WRITE(*,*)’PRINT X-Y DATA TO FILE "xy.dat" ? :’
      READ(*,*) LOG1
      If(LOG1.NE.’n’) OPEN(UNIT=10, FILE=’xy.dat’)
      WRITE(*,*)’PRINT X-Z DATA TO FILE "xz.dat" ? :’
      READ(*,*) LOG2
      If(LOG2.NE.’n’) OPEN(UNIT=20, FILE=’xz.dat’)
      WRITE(*,*)’PRINT Y-Z DATA TO FILE "yz.dat" ? :’
      READ(*,*) LOG3
      If(LOG3.NE.’n’) OPEN(UNIT=30, FILE=’yz.dat’)
      WRITE(*,*)’PRINT X-Y-Z DATA TO FILE "xyz.dat" ? :’
      READ(*,*) LOG4
      If(LOG4.NE.’n’) OPEN(UNIT=40, FILE=’xyz.dat’)
      WRITE(*,*)’PRINT X-T DATA TO FILE "xt.dat" ? :’
      READ(*,*) LOG5
      IF(LOG5.NE.’n’) OPEN(UNIT=50, FILE=’xt.dat’)

      WRITE(*,*)’@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@’
      WRITE(*,*)’ program is running ...’

C---------------- MAINPART : ---------------------------

      DO WHILE(T.LE.T_END)

        STP = STP + 1

        IF(LOG_A.EQ..TRUE.) THEN
          IF(T.GT.T_UP) THEN
               STP_UP = STP
              LOG_A = .FALSE.
          END IF
        END IF

C--------------- WRITING DATAS TO FILE : ---------------------

          IF (T.GE.(COUNTER * T_PRINT + T_UP)) THEN
              IF(LOG1.NE.’n’) WRITE(10,*) S(1),S(2)
              IF(LOG2.NE.’n’) WRITE(20,*) S(1),S(3)
              IF(LOG3.NE.’n’) WRITE(30,*) S(2),S(3)
              IF(LOG4.NE.’n’) WRITE(40,*) S(1),S(2),S(3)
              IF(LOG5.NE.’n’) WRITE(50,*) T, S(1)

              COUNTER = COUNTER + 1
              WRITE(*,*) COUNTER

           END IF

C------------------- INTEGRATION SCHEME : --------------------
        T_COPY = T
          EST1 = S
          EST2 = S

        CALL DRKSTP(3,H,T,EST1,RHS,DSDT1)

             T = T_COPY

        CALL DRKSTP(3,H/2.0d0,T,EST2,RHS,DSDT2)
        CALL DRKSTP(3,H/2.0d0,T,EST2,RHS,DSDT2)

C---------- COMPUTE ESTIMATES AND RESULTING ERROR : ---------------

         EST(1) = DABS( EST1(1)-EST2(1) )
         EST(2) = DABS( EST1(2)-EST2(2) )
         EST(3) = DABS( EST1(3)-EST2(3) )
          ERROR = EST(1)
         IF (EST(2).GT.ERROR) ERROR = EST(2)
         IF (EST(3).GT.ERROR) ERROR = EST(3)

C------------ ADAPTIVE STEPSIZE : ------------------------
C-------- DETERMING NEW STEPSIZE H : ---------------------

        IF (ERROR.GT.TOLERANCE) THEN
           H_NEW = H * (TOLERANCE/ERROR)**.25d0
               T = T - H
        END IF

        IF (ERROR.LE.TOLERANCE) THEN

          IF (ERROR.EQ.0.0d0) THEN
            H_NEW = 5.0d0 * H
          ELSE
            H_NEW = H * (TOLERANCE/ERROR)**.2d0
          END IF

          S = (16.0d0 * EST2 - EST1) / 15.0d0
        END IF

       IF (H_NEW.LT.(H/5.0d0)) then
         H_NEW = (H/5.0d0)
       END IF
       IF (H_NEW.GT.(5.0d0*H)) THEN
         H_NEW = (5.0d0*H)
       END IF

       IF (H_NEW.LT.(1.0d-15)) THEN
         H_NEW = 1.0d-15
       END IF

       H = H_NEW

      END DO

      PRINT*,’...end of running !!!’

C------------------ LAST SCREEN OUTPUT : -----------------------

      WRITE(*,*)’@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@’
      WRITE(*,*)’YOUR CHOSEN CONFIGURATION WAS :’
      WRITE(*,*)’ ’
      WRITE(*,*)’# TOLERANCE = ’,TOLERANCE
      WRITE(*,*)’ ’
      WRITE(*,*)’# DATA ARE PRINTED TO FILE BETWEEN :’
      WRITE(*,*)’ TIME T-UP = ’,T_UP,’ ( STEP-NR ’,STP_UP,’ )’

      WRITE(*,*)’ AND’
      WRITE(*,*)’ TIME T_END = ’,T_END,’ ( STEP-NR ’,STP,’ )’
      WRITE(*,*)’# NUMBER OF WRITTEN POINTS = ’,COUNTER,’ OF ’,NR_DATA
      WRITE(*,*)’ ( DISTANCE OF DATA POINTS = CONST = ’,T_PRINT,’ )’
      WRITE(*,*)’ ’
      WRITE(*,*)’ # PARAMETER : SIGMA = ’,SIGMA
      WRITE(*,*)’ R = ’,R
      WRITE(*,*)’ B = ’,B
      WRITE(*,*)’ AX = ’,AX
      WRITE(*,*)’ AY = ’,AY
      WRITE(*,*)’ AZ = ’,AZ
      WRITE(*,*)’# INITIAL CONFIGURATION : Xo = ’,So(1)
      WRITE(*,*)’ Yo = ’,So(2)
      WRITE(*,*)’ Zo = ’,So(3)
      WRITE(*,*)’@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@’

      STOP
      END

C--------------------------------------------------------------------------

c******************************************************
c*** SUBROUTINE RHS ***
c*** COMPUTES THE RIGHT HAND SIDE OF THE EQUATIONS ***
c******************************************************
      SUBROUTINE RHS(T, S, DSDT)

      IMPLICIT NONE
      REAL*8 S(3), DSDT(3)
      REAL*8 SIGMA, R, B, AX, AY, AZ, T
      COMMON SIGMA, R, B, AX, AY, AZ

      DSDT(1) = - SIGMA * S(1) + SIGMA * S(2) + AX
      DSDT(2) = - S(1) * S(3) + R * S(1) - S(2) + AY
      DSDT(3) = S(1) * S(2) - B * S(3) + AZ

      RETURN
      END

C----------------------------------------------------------------------------